/*
CpGears v1.0

Copyright (c) 2010 Yves Riel http://www.flash-factor.com

Permission is hereby granted, free of charge, to any person obtaining a copy
of this software and associated documentation files (the "Software"), to deal
in the Software without restriction, including without limitation the rights
to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
copies of the Software, and to permit persons to whom the Software is
furnished to do so, subject to the following conditions:

The above copyright notice and this permission notice shall be included in
all copies or substantial portions of the Software.

THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN
THE SOFTWARE.
*/
package cpgears.helpers {
	import flash.display.DisplayObject;
	import flash.display.DisplayObjectContainer;
	import flash.utils.describeType;
	
	/**
	 * Helper class to manipulate DisplayObjects.
	 * 
	 * @author Whyves
	 */
	public class DisplayObjectUtil{
		
		/**
		 * Finds the display object parent's that matched the provided name.
		 * 
		 * @param	source The display object from which to start searching.
		 * @param	parentName The name of the parent (case sensitive).
		 * @return  The parent instance or <code>null</code> if no parent was found.
		 */
		public static function findParentByName(source:DisplayObject, parentName:String):DisplayObjectContainer {
			var currentTarget:DisplayObjectContainer = source.parent;
			while (currentTarget != null) {
				if (currentTarget.name.indexOf(parentName) >= 0) return currentTarget;
				currentTarget = currentTarget.parent;
			}
			return null; // We're at the root
		}
		
		/**
		 * Finds the display object parent's that matched the provided class type.
		 * 
		 * @param	source The display object from which to start searching.
		 * @param	parentType The type of the parent (case sensitive).
		 * @return  The parent instance or <code>null</code> if no parent was found.
		 */
		public static function findParentByType(source:DisplayObject, parentType:String):DisplayObjectContainer {
			var currentTarget:DisplayObjectContainer = source.parent;
			while (currentTarget != null) {
				if (describeType(currentTarget).@name.indexOf(parentType) >= 0) return currentTarget;
				currentTarget = currentTarget.parent;
			}
			return null; // We're at the root
		}
	}
}